
Abstract. The combination of probabilistic and nondeterministic choice
in program calculi is a notoriously tricky problem, and one with a long
history. We present a simple functional programming approach to this
challenge, based on algebraic theories of computational effects. We make
use of the powerful abstraction facilities of modern functional languages,
to introduce the choice operations as a little embedded domain-specific
language rather than having to define a language extension; we rely on
referential transparency, to justify straightforward equational reasoning
about program behaviour.

For Carroll Morgan, on his sixtieth birthday.

1 Introduction

In the preface of his book [31] with Annabelle McIver on “rigorous reasoning
about probabilistic programs and systems”, Carroll Morgan writes:

Economy and precision of reasoning are what we have come to expect for
standard programs; there is no reason we should accept less when they are
probabilistic.

The book goes on to present the probabilistic Guarded Command Language
pGCL, an extension of Dijkstra’s GCL [6] to include probabilistic choice as well
as the usual nondeterministic choice, and its semantics in terms of expectation
transformers, the generalization of predicate transformers from boolean predi-
cates to numeric expectations.

In this paper, we present an alternative approach to rigorous reasoning about
probabilistic programs and systems, also offering the desired economy and pre-
cision. Our approach is based on pure functional programming, rather than on
Dijkstra’s imperative style. The main advantage of pure functional programming
when it comes to manipulating programs is its referential transparency, which
means that familiar equational reasoning works as well for expressions denoting
programs as it does for expressions denoting numbers. As a consequence, we no
longer need two distinct domains of discourse—a programming notation in which

to express computations, and a logic in which to reason about them—because
the one language works for both.

A second advantage of functional programming is the facilities it provides
for defining new abstractions ‘within the language’, rather than requiring one to
step out into the meta-language in order to define a new language. Our chosen
language Haskell does not provide constructs for probabilistic choice itself, but
that is no obstacle: instead, it provides the necessary abstraction mechanisms
that allow us to define those constructs ourselves. Rather than a new language
‘probabilistic Haskell’, we can define probabilistic choice within standard Haskell;
one might characterize the result as an embedded domain-specific language for
probabilistic programming.

The apparent drawback of pure functional programming is that its very pu-
rity appears to rule out interesting computational effects, not least probabilistic
choice. After all, if coin denotes the computation modelling a fair coin toss—
a 50–50 choice between heads and tails—then do not two occurrences of coin
denote possibly different outcomes, thereby destroying referential transparency?
The apparent problem is eliminated by distinguishing between types that repre-
sent values, such as ‘heads’, and those that represent computations, such as coin.
Two occurrences of coin denote the same computation, and it is the execution
of these computations that may yield different outcomes. Each class of effects,
such as probabilistic choice, determines a notion of computation, in this case of
probability distributions; coin denotes not a single outcome, but a distribution
of outcomes. The operations and axioms of a notion of computation can be pre-
cisely and elegantly abstracted as a monad, or equivalently an algebraic theory ;
equational reasoning can be conducted within such a theory.

The paper is intended as a tutorial and a manifesto, rather than presenting
any new results. We start with a brief introduction to pure functional program-
ming and to the use of monads to capture computational effects (Section 2)—
readers familiar with functional programming in general, and Haskell in partic-
ular, may wish to skip this section. We then introduce theories of nondetermin-
istic choice (Section 3) and probabilistic choice (Section 4) separately, and in
combination (Section 5). Section 6 presents an extended example based on the
infamous Monty Hall problem. In Section 7 we consider the possibility of failure
and the effect of exceptions, which gives rise to conditionally probabilistic com-
putations; and in Section 8 we look at recursive definitions. Section 9 concludes
with a discussion of related work and some thoughts about future developments.

2 Effectful functional programming

Pure functional programming languages constitute a very appealing model of
computation: simple, due to abstraction from the details of computer architec-
ture, yet still expressive, allowing concise specification of complex constructions.
These strengths derive from referentially transparency : as far as the semantics is
concerned, the only relevant aspect of any expression is the value it denotes. In
particular, expressions have no side-effects; so any subexpression can be replaced

by any other having the same value, without affecting the surrounding context.
Expressions therefore behave like ordinary high-school algebra, and reasoning
about programs can be conducted using ordinary high-school equational reason-
ing, substituting one subexpression for another with equal value. Consequently,
one’s language for programming is simultaneously one’s language for reason-
ing about those programs—there is no need to step outside the programming
notation to a different logical domain such as predicate calculus.

2.1 Functional programming

The essence of functional programming is that programs are equations and func-
tions are values. For example, the squaring function on integers might be defined:

square :: Int → Int
square x = x × x

or equivalently

square :: Int → Int
square = λx → x × x

As well as specifying an action, namely how to compute squares, this program
also serves as an equation: for any x , the expression square x is equivalent to the
expression x × x , and either may be replaced anywhere by the other (taking due
care over bound variables); similarly, the identifier square itself may be replaced
by the lambda expression λx → x × x denoting the squaring function. Likewise,
function composition (◦) is a value, just like any other, albeit a higher-order one:

(◦) :: (b → c)→ (a → b)→ a → c
(f ◦ g) x = f (g x )

Functional programmers restrict themselves to manipulating expressions,
rather than statements. So in order to regain the expressivity provided by state-
ments in imperative languages, functional programming must provide an en-
riched expression language. Higher-order operators like functional composition
go some way towards this. Another powerful tool is to allow complex data struc-
tures to be denotable as expressions; for example, the datatype [a ] of lists of
elements each of type a might be defined as follows:

data [a ] = [ ] | a : [a ]

With this device, a data structure such as the list of three elements 1 : (2 : (3 : [ ]))
can be denoted as an expression; in contrast, in conventional imperative lan-
guages, complex data structures such as lists and trees can generally be con-
structed only via a sequence of side-effecting assignment statements.

Functions over such data structures can conveniently be defined by pattern
matching. For example, here is the standard function foldr to fold a list to a
value:

foldr :: (a → b → b)→ b → [a ]→ b
foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs)

This is another higher-order function, since it takes a function as its first argu-
ment. One instance of foldr is the function sum that sums a list of integers:

sum :: [Int ]→ Int
sum = foldr (+) 0

Another is the higher-order function map that applies an argument f to each
element of a list:

map :: (a → b)→ ([a ]→ [b ])
map g = foldr (λx ys → g x : ys) [ ]

Lists are a polymorphic datatype; the polymorphism is expressed precisely by
map. Polymorphic functions such as reverse :: [a ] → [a ] are those that de-
pend only on the structure of a datatype, oblivious to the elements; their poly-
morphism is expressed precisely by a corresponding naturality property, stating
that they commute with the appropriate map function—for example, reverse ◦
map f = map f ◦ reverse [52].

2.2 Equational reasoning

Referential transparency means that plain ordinary equational reasoning suffices
for proving properties of programs. For example, one very important property
of the foldr function is the fusion law :

h ◦ foldr f e = foldr f ′ e ′ ⇐= h (f x y) = f ′ x (h y) ∧ h e = e ′

One way of proving this law is by induction over lists (which we assume here to
be finite). For the base case, we have:

h (foldr f e [ ])
= [[ definition of foldr ]]
h e

= [[ assumption ]]
e ′

= [[ definition of foldr ]]
foldr f ′ e ′ [ ]

For the inductive step, we assume that the result holds on xs, and calculate for
x : xs as follows:

h (foldr f e (x : xs))
= [[ definition of foldr ]]
h (f x (foldr f e xs))

= [[ assumption ]]
f ′ x (h (foldr f e xs))

= [[ inductive hypothesis ]]
f ′ x (foldr f ′ e ′ xs)

= [[ definition of foldr ]]
foldr f ′ e ′ (x : xs)

A simple consequence of the fusion law is the fold–map fusion law, when h is
itself an instance of foldr , and follows a map over lists, which is another instance
of foldr . In this case, the fusion result

foldr f e ◦map g = foldr f ′ e ′

follows from the fusion conditions

foldr f e (g x : ys) = f ′ x (foldr f e ys) ∧ foldr f e [ ] = e ′

These in turn are satisfied if e ′ = e and f ′ = λx z → f (g x ) z = f ◦ g ; that is,

foldr f e ◦map g = foldr (f ◦ g) e

For most of the paper we will assume a setting of total functions between sets.
In this setting, arbitrary recursive definitions do not in general admit canoni-
cal solutions; we restrict attention to well-founded recursions such as that in
foldr , and correspondingly to finite data structures. We only have to relax this
restriction in Section 8.

2.3 Effects in pure functional languages

Equational reasoning about pure computations is all very well, but to be use-
ful, computations must have some observable effects. It may seem at first that
equational reasoning must then be abandoned. After all, as soon as one al-
lows state-mutating statements such as x := x + 1 in a programming language,
the high-school algebra approach to reasoning no longer works; and similarly
for other classes of effect, such as input/output, nondeterminism, probabilistic
choice, exceptions, and so on.

Moggi [36] famously showed how the well-understood concept of a monad
from category theory provides exactly the right interface to an abstraction of
computational effects such as mutable state, allowing the development of an
elegant yet expressive computational lambda calculus for modelling programming
languages with effects. Wadler [54] popularized this approach within functional
programming, and it quickly became the technology of choice for integrating
effects into lazy functional programming languages such as Haskell.

With the monadic approach to computational effects, purely functional ex-
pressions are classifed into two kinds: those that denote values like integers and
strings, and those that denote computations with possible effects. However, both
are represented as pure data—the computations are represented as pure terms

in a certain abstract syntax, rather than some kind of impure action. When the
run-time system of a language encounters the first kind of expression, it evaluates
it and prints it out; when it encounters the second kind, it evaluates it, interprets
the term as the effectful computation it encodes, and executes that computa-
tion. Consequently, evaluation remains pure, and any impurities are quarantined
within the run-time system.

The abstract syntax needed to capture effectful computations is very simple.
There is a general framework consisting of just two operators, which in a sense
model the compositional structure of computations; then for each class of effect,
there is an extension to the general framework to model the primitives specific
to that class. (In fact, the general framework and a specific extension together
represent the free term algebra for the signature corresponding to the primitives
for a particular class of effects. It is no coincidence that monads turn out to be
useful for modelling such term algebras, because they were developed precisely
as a categorical expression of universal algebra [30]. We return to this point in
Section 9.3.)

The general framework can be expressed as a type class in Haskell:

class Monad m where
return :: a → m a
(>>=) :: m a → (a → m b)→ m b
fmap :: (a → b)→ (m a → m b)
join :: m (m a)→ m a

p >>= k = join (fmap k p)
join pp = pp >>= id
fmap f p = p >>= (return ◦ f )

This declaration states that the type constructor (that is, operation on types)
m is in the type class Monad if we can provide suitable definitions of the four
methods return, (>>=), fmap, and join, with the given types. In fact, the methods
are interdefinable, and some have default definitions in terms of others; it is
necessary to define return, but it suffices to define either (>>=) or both fmap
and join. (We have chosen this presentation allowing alternative definitions for
flexibility; it is different from but equivalent to the Monad class in the Haskell
standard libraries.)

Technically, the methods should also satisfy some laws, although these cannot
be stated in the Haskell type class declaration:

return x >>= k = k x -- left unit
p >>= return = p -- right unit
(p >>= k)>>= k ′ = p >>= (λx → k x >>= k ′) -- associativity

fmap id = id -- map–identity
fmap (f ◦ g) = fmap f ◦ fmap g -- map–composition

join ◦ return = id -- left unit
join ◦ fmap return = id -- right unit
join ◦ fmap join = join ◦ join -- associativity

(Throughout the paper, we make the following naming conventions: p, q , r de-
note monadic terms or ‘programs’, h, k denote functions yielding monadic terms,
x , y , z denote polymorphic variables, a, b, c denote booleans, l ,m,n denote inte-
gers, and u, v ,w denote probabilities.) Informally, the type m a denotes compu-
tations that may have some effect, and that yield results of type a. The function
return lifts plain values into pure computations. The operator >>=, pronounced
‘bind’, acts as a kind of sequential composition; the second computation may
depend on the result returned by the first, and the overall result is the result of
the second. The first three laws can be seen as unit and associativity properties
of this form of sequential composition. The function join flattens a computation
that yields computations that yield results into a computation that yields results
directly, and the function fmap modifies each of the results of a computation;
together with return, these two give an alternative (equivalent) perspective on
sequential composition.

Two shorthands turn out to be quite convenient. We write skip for the pure
computation that returns the unit ():

skip :: Monad m ⇒ m ()
skip = return ()

and >> for the special case of >>= in which the second computation is independent
of the result returned by the first:

(>>) :: Monad m ⇒ m a → m b → m b
p >> q = p >>= (λ → q)

These two shorthands more obviously form analogues of the ‘skip’ and ‘sequential
composition’ operators of imperative programming languages. For example, with
these we can form the sequential composition of a sequence of unit-returning
computations, discarding all the unit results and returning unit overall. (This
is actually a type specialization of the corresponding function in the Haskell
standard library, but it is sufficient for our purposes.)

sequence :: Monad m ⇒ [m ()]→ m ()
sequence = foldr (>>) skip

This function reveals one of the beauties of pure and lazy functional program-
ming: if a useful control construct is missing from the language, it is usually
possible to define it as an ordinary value rather than having to extend the syntax
and the compiler. Another famous example is the conditional; if Haskell didn’t
already provide the if ... then ... else... construct, something entirely equivalent
(except for the concrete syntax) could be defined—the same cannot be said of a
language providing only eager evaluation. And because conditional would be an
ordinary value, the ordinary principles of reasoning would apply; for example,
function application distributes leftwards and rightwards over conditional:

f (if b then x else y) = if b then f x else f y
(if b then f else g) x = if b then f x else g x

These laws can easily be verified by considering the two cases b = True and
b = False. (In fact, the first law as stated only holds in SET. Once one moves to
CPO, one also needs to consider the case that b is undefined; then the first law
only holds when f is strict. The second law is still unconditional, provided that
λx → ⊥ = ⊥; this is the case with flat function spaces, the usual presentation
in CPO, but not in fact in Haskell with the seq operator, which distinguishes
between ⊥ and λx → ⊥.) In particular, letting f be (>>=k) and (p>>=), we
deduce from the first law that composition distributes respectively leftwards
and rightwards over conditional:

(if b then p else q)>>= k = if b then p >>= k else q >>= k
p >>= (if b then k else k ′) = if b then p >>= k else p >>= k ′

(Again, these laws hold unconditionally in SET; in CPO, they require >>= to be
strict in its left and right argument, respectively.)

2.4 State

So much for the general framework; here is an extension to capture mutable state
as a class of effects. Just two additional operations are required: get , to read the
state, and put , to update it. We declare a subclass MonadState of Monad ; type
constructor m is a member of the class MonadState if it is a member of Monad
and it supports the two additional methods get and put . (To be precise, the
subclass is MonadState s for some fixed state type s, and it encompasses type
constructors m that support mutable state of type s; the vertical bar precedes
a ‘functional dependency’, indicating that m determines s.)

class Monad m ⇒ MonadState s m | m → s where
get :: m s
put :: s → m ()

As with the two methods of the Monad interface, it is not sufficient simply to
provide implementations of get and put that have the right types—they should
also satisfy some laws:

get >>= λs → get >>= λs ′ → k s s ′ = get >>= λs → k s s -- get–get
get >>= put = skip -- get–put
put s >> put s ′ = put s ′ -- put–put
put s >> get >>= λs ′ → k s ′ = put s >> k s -- put–get

Informally: two consecutive gets will read the same value twice; getting a value
then putting it back is has no effect; two consecutive puts are equivalent to just
the second one; and a get immediately after putting a value will yield that value.

For example, here is a simple expression denoting a computation on a mutable
integer state, which reads the current state, increments it, writes the new value
back, and then returns the parity of the original value.

incrodd :: MonadState Int m ⇒ m Bool
incrodd = get >>= (λn → put (n + 1)>>= (λ()→ return (odd n)))

There is an obvious simulation of mutable state in terms of state-transforming
functions. A computation that acts on a state of type s, and yields a result of
type a, can be represented as a function of type s → (a, s):

type State s a = s → (a, s)

Now, State s forms a type of computations, and so can be made an instance
of the type class Monad—for return a we use the state-transforming function
that yields x and leaves the state unchanged; fmap f applies f to the output
value without touching the output state; and join collapses a state-transformer
that yields a state-transformer by applying the output state-transformer to the
output state:

instance Monad (State s) where
return x = λs → (x , s)
fmap f p = λs → let (x , s ′) = p s in (f x , s ′)
join p = λs → let (p′, s ′) = p s in p′ s ′

The reader may enjoy deriving from this the corresponding definition p >>= k =
λs → let (x , s ′) = p s in k x s ′ of bind, which chains state transformations
together. (For technical reasons, this instance declaration is not quite in Haskell
syntax: rather than a type synonym, State ought to be a newtype or datatype,
with a constructor and deconstructor. But what is shown is morally correct.)

Of course, State s supports the features of mutable state—get yields a copy
of the state, and put overwrites it:

instance MonadState s (State s) where
get = λs → (s, s)
put s ′ = λs → ((), s ′)

As it happens, the datatype State s is (isomorphic to) the free term algebra on
the MonadState s signature, modulo the four laws of get and put [42].

2.5 Imperative functional programming

Wadler also observed [53] that the methods of the Monad interface are sufficient
to implement a notation based on the set comprehensions of Zermelo–Fraenkel
set theory. This too has found its way into Haskell, as the ‘do notation’ [25],
which is defined by translation into Monad methods as follows:

do {p} = p
do {x ← p ; qs } = p >>= λx → do {qs }
do {p ; qs } = p >> do {qs }
do {let decls ; qs } = let decls in do {qs }

In particular, instead of having to write functions (typically lambda expressions)
as the second argument of >>=, with the do notation we can write a generator
x ← p to bind a new variable x that is in scope in all subsequent qualifiers.
Using this notation, we can rewrite the incrodd program above as follows:

incrodd :: MonadState Int m ⇒ m Bool
incrodd = do {n ← get ; put (n + 1) ; return (odd n)}

The three monad laws appear in the do notation as follows:

do {x ← return e ; k x } = do {k e }
do {x ← p ; return x } = do {p}
do {x ← p ; y ← h x ; k y } = do {y ← do {x ← p ; h x } ; k y }

(where, implicitly in the third law, x is not free in k). Distribution of composition
leftwards and rightwards over conditional looks like this:

do {x ← if b then p else q ; k x } = if b then do {x ← p ; k x }
else do {x ← q ; k x }

do {x ← p ; if b then h x else k x } = if b then do {x ← p ; h x }
else do {x ← p ; k x }

(where, implicitly in the second law, x is not free in b). The four laws of state
become:

do {s ← get ; s ′ ← get ; k s s ′} = do {s ← get ; k s s } -- get–get
do {s ← get ; put s } = do {skip} -- get–put
do {put s ; put s ′} = do {put s ′} -- put–put
do {put s ; s ′ ← get ; k s ′} = do {put s ; k s } -- put–get

The do notation yields a natural imperative programming style, as we hope the
rest of this paper demonstrates; indeed, it has been said that “Haskell is the
world’s finest imperative programming language” [40].

2.6 An example of simple monadic equational reasoning

To summarize: the Monad class provides an interface for sequencing computa-
tions; one should program to that interface where appropriate, making subclasses
of Monad for each specific class of effects; and the interface ought to specify
laws as well as signatures for its methods. We have recently argued [10] that
this perspective on monads is precisely the right one for equational reasoning
about effectful programs—contrary to popular opinion, the impurities of com-
putational effects offer no insurmountable obstacles to program calculation, at
least when they are properly encapsulated. To illustrate this claim, we present
a simple example of reasoning with stateful computations.

Here is a simple stateful computation to add an integer argument to an
integer state:

add :: MonadState Int m ⇒ Int → m ()
add n = do {m ← get ; put (m + n)}

We claim that adding each integer in a list in sequence to an integer state is the
same as adding their sum all at once:

addAll = add ◦ sum

where addAll turns each integer in a list into an integer-adding computation,
then sequences this list of computations:

addAll :: MonadState Int m ⇒ [Int ]→ m ()
addAll = sequence ◦map add

Because sequence is an instance of foldr , we can combine the two phases of
addAll into one, using the fold–map fusion law:

addAll = foldr (λn p → do {add n ; p}) skip

Now, since sum and addAll are both instances of foldr , the claim is an instance
of the standard fusion law, and follows from two simple fusion properties:

add 0 = skip
add (n + n ′) = do {add n ; add n ′}

For the first of these, we have:

add 0
= [[ add ]]

do { l ← get ; put (l + 0)}
= [[ arithmetic ]]

do { l ← get ; put l }
= [[ get–put ]]
skip

And for the second, starting from the more complicated right-hand side, we have:

do {add n ; add n ′}
= [[ add ]]

do {do {m ← get ; put (m + n)} ; do { l ← get ; put (l + n ′)}}
= [[ associativity ]]

do {m ← get ; put (m + n) ; l ← get ; put (l + n ′)}
= [[ put–get ]]

do {m ← get ; put (m + n) ; put ((m + n) + n ′)}
= [[ associativity of addition ]]

do {m ← get ; put (m + n) ; put (m + (n + n ′))}
= [[ put–put ]]

do {m ← get ; put (m + (n + n ′))}

= [[ add ]]
add (n + n ′)

which completes the proof.
Of course, sum and addAll are two rather special functions, both being in-

stances of the easily manipulated foldr pattern. However, that is incidental to
our point: if we had picked an example involving a more complicated pattern of
computation, then the reasoning would certainly have been more complicated
too, but it would still have been plain ordinary equational reasoning—reasoning
about the computational effects would pose no more of a problem.

3 An algebraic theory of nondeterministic choice

Let us now turn to a different class of effects. Nondeterministic programs are
characterized by the ability to choose between multiple results. We model this
as a subclass of Monad .

class Monad m ⇒ MonadAlt m where
(�) :: m a → m a → m a

We stipulate that � is associative, commutative, and idempotent:

(p � q) � r = p � (q � r)
p � q = q � p
p � p = p

and that composition distributes leftwards over it:

do {x ← (p � q) ; k x } = do {x ← p ; k x } � do {x ← q ; k x }

However, we do not insist that composition distributes rightwards over choice:
in general,

do {x ← p ; (h x � k x )} 6= do {x ← p ; h x } � do {x ← p ; k x }

This is in order to accommodate both angelic and demonic interpretations of
nondeterminism. One distinction between the two is in terms of the number
of branches of a choice that an implementation might choose to follow: angelic
choice will explore both branches, whereas demonic choice is free to pick either
branch but will not follow both. In particular, consider the case that computa-
tion p has some non-idempotent effects in addition to nondeterminism, such as
writing output. If � is angelic, then these effects happen once on the left-hand
side of the equation, and twice on the right; whereas if � is demonic, just one
branch of each choice will be picked, and the two sides of the equation are indeed
equal.

On account of the associativity, commutativity, and idempotence of choice,
the essential—indeed, the initial, in the categorical sense—semantics of a nonde-
terministic computation amounts to a finite nonempty set of alternative results.

In other words, we can simulate a computation that exploits just the effect of
choice as a function that returns a nonempty set of results. A pure computation
amounts to returning a singleton set, fmap f applies f to each element of a set,
and a computation of computations can be flattened by taking the union of the
resulting set of sets. (The operational behaviour of an implementation will differ,
depending on the interpretation of choice: an angelic implementation will deliver
the whole set of results; a demonic implementation will pick one arbitrarily. But
either way, the semantics is represented as a function yielding a set.)

A convenient approximate implementation of finite nonempty sets is in terms
of nonempty lists—‘approximate’ in the sense that we consider two lists to rep-
resent the same set of results if they are equal up to reordering and duplication
of elements.

instance Monad [ ] where
return a = [a ]
fmap f p = [f x | x ← p ]
join = concat

Naturally, we implement the nondeterministic choice as concatenation:

instance MonadAlt [ ] where
(�) = (++)

In some other contexts, we might not want such a strong collection of laws for
nondeterministic choice. For example, if we are modelling search strategies [14,
47], we might want to treat as significant the order in which results are found,
and so we might want to drop the commutativity axiom.

3.1 Example: subsequences of a list

As an example of reasoning with nondeterministic programs, here is a rendition
in terms of choice of the function subs that nondeterministically chooses a sub-
sequence of a list. Of course, interpreted in the nonempty-list implementation of
nondeterminism, subs returns the usual nonempty list of lists; but this definition
supports other implementations of nondeterminism too, such as bags and sets.

subs :: MonadAlt m ⇒ [a ]→ m [a ]
subs [ ] = return [ ]
subs (x : xs) = fmap (x :) xss � xss where xss = subs xs

Here is a simple property that we might wish to prove—that subs distributes
over list concatenation:

subs (xs ++ ys) = do {us ← subs xs ; vs ← subs ys ; return (us ++ vs)}

Using the laws of nondeterminism, this property of an effectful program can be
proved by induction over xs, using plain ordinary equational reasoning. For the
base case xs = [ ], we have:

do {us ← subs [ ] ; vs ← subs ys ; return (us ++ vs)}
= [[ definition of subs ]]

do {us ← return [ ] ; vs ← subs ys ; return (us ++ vs)}
= [[ left unit ]]

do {vs ← subs ys ; return ([ ] ++ vs)}
= [[ definition of ++ ]]

do {vs ← subs ys ; return vs }
= [[ right unit ]]
subs ys

= [[ by assumption, xs = [ ] ]]
subs (xs ++ ys)

For the inductive step, we assume the result for xs, and calculate for x : xs as
follows:

do {us ← subs (x : xs) ; vs ← subs ys ; return (us ++ vs)}
= [[ definition of subs; let xss = subs xs ]]

do {us ← (fmap (x :) xss � xss) ; vs ← subs ys ; return (us ++ vs)}
= [[ composition distributes leftwards over � ]]

do {us ← fmap (x :) xss ; vs ← subs ys ; return (us ++ vs)} �
do {us ← xss ; vs ← subs ys ; return (us ++ vs)}

= [[ fmap and do notation ]]
do {us ′ ← xss ; vs ← subs ys ; return ((x : us ′) ++ vs)} �
do {us ← xss ; vs ← subs ys ; return (us ++ vs)}

= [[ definition of ++; do notation ]]
fmap (x :) (do {us ′ ← xss ; vs ← subs ys ; return (us ′ ++ vs)}) �
do {us ← xss ; vs ← subs ys ; return (us ++ vs)}

= [[ by assumption, xss = subs xs; inductive hypothesis, twice ]]
fmap (x :) (subs (xs ++ ys)) � subs (xs ++ ys)

= [[ definition of subs ]]
subs (x : (xs ++ ys))

= [[ definition of ++ ]]
subs ((x : xs) ++ ys)

Again, plain ordinary equational reasoning suffices, using programs as equations
and the axioms of nondeterminism.

4 An algebraic theory of probabilistic choice

Here is another class of effects. Probabilistic computations are characterized by
the ability to make a probabilistic choice between alternatives. We suppose a type
Prob of probabilities (say, the rationals in the closed unit interval), and define
a Monad subclass for computations drawing from finitely supported probability
distributions, that is, distributions in which only a finite number of elements
have positive probabilities:

class Monad m ⇒ MonadProb m where
choice :: Prob → m a → m a → m a

The idea is that choice w p q behaves as p with probability w and as q with
probability 1−w . From now on, we will write ‘w ’ for 1−w , and following Hoare’s
and Morgan’s convention [15, 17], write choice in infix notation, ‘p/w.q ’, because
this makes the laws more legible. We have two identity laws:

p / 0 . q = q
p / 1 . q = p

a quasi-commutativity law:

p / w . q = q / w . p

idempotence:

p / w . p = p

and quasi-associativity:

p / u . (q / v . r) = (p / w . q) / x . r ⇐= u = w × x ∧ x = u × v

As informal justification for quasi-associativity, observe that the likelihoods of
p, q , r on the left are u, u × v , u × v , and on the right are w × x ,w × x , x , and a
little algebra shows that these are pairwise equal, given the premise.

As a final pair of laws, we stipulate that bind distributes both leftwards and
rightwards over choice:

do {x ← (p / w . q) ; k x } = do {x ← p ; k x } / w . do {x ← q ; k x }
do {x ← p ; (h x / w . k x ) = do {x ← p ; h x } / w . do {x ← p ; k x }

where, in the second law, x is assumed not to occur free in w . (In contrast to
nondeterministic choice, we have both distributivities here. This means that, op-
erationally, an implementation may take either branch of a probabilistic choice,
but not both—like demonic choice, and unlike angelic.)

For example, a fair coin can be modelled as a 50–50 probabilistic choice
between heads and tails (represented as booleans here):

coin :: MonadProb m ⇒ m Bool
coin = return True / 1/2 . return False

One obvious representation to pick as an implementation of MonadProb uses
probability-weighted lists of values; thus, coin might be represented as the list
[(True, 1/2), (False, 1/2)].

type Dist a = [(a,Prob)] -- weights sum to 1

A pure computation is represented as a point distribution, mapping applies a
function to each element, and a distribution of distributions can be flattened by
taking a kind of weighted cartesian product:

instance Monad Dist where
return x = [(x , 1)]
fmap f p = [(f x ,w) | (x ,w)← p ]
join p = concat [scale w x | (x ,w)← p ]

where

scale :: Prob → [(a,Prob)]→ [(a,Prob)]
scale v p = [(x , v × w) | (x ,w)← p ]

On the other hand, / . is a kind of weighted sum:

instance MonadProb Dist where
p / w . q = scale w p ++ scale w q

Probability-weighted lists are not quite the initial model, because the identity,
idempotence, quasi-commutativity, and quasi-associativity laws of / . do not
hold. In fact, the initial model of the specification consists of finite mappings
from elements to probabilities, collected from these weighted lists in the obvious
way—at least, for an element type in the type class Eq , supporting the equality
operation needed by finite maps, we can define:

collect :: Eq a ⇒ Dist a → (a → Prob)
collect p y = sum [w | (x ,w)← p, x y ]

That is, equivalences on Dist ought to be taken modulo permutations, zero-
weighted elements, and repeated elements (whose weights should be added).
Nevertheless, the datatype Dist itself provides a convenient approximation to
the initial model.

Quasi-associativity can make the arithmetic of weights rather complicated,
especially when choices are nested. Inspired by Morgan’s distribution compre-
hensions [38], we sometimes make use of a flat notation for nested choices. For
example, instead of (p / 1/2 . q) / 1/3 . (r / 1/4 . s) we allow ourselves to write
〈p@1/6, q@1/6, r@1/6, s@1/2〉, multiplying out all the probabilities.

4.1 Example: uniform distributions

Extending the fair coin example, we might define uniform distributions

uniform :: MonadProb m ⇒ [a ]→ m a -- nonempty list
uniform [x ] = return x
uniform (x : xs) = return x / 1/length (x :xs) . uniform xs

so that coin = uniform [True,False ], and uniform [1, 2, 3] = return 1 / 1/3 .
(return 2 / 1/2 . return 3).

Choices drawn from uniform distributions but never used are free of side-
effects, and so can be discarded: it is a straightforward proof by induction over
xs that

do {x ← uniform xs ; p} = p

when p does not depend on x . Similarly, uniform distributes over concatenation:

uniform (xs ++ ys) = uniform xs / m/m+n . uniform ys

where m = length xs and n = length ys. As a consequence of these proper-
ties of uniform, we can conclude that consecutive choices drawn from uniform
distributions are independent; that is, choosing consecutively from two uniform
distributions is equivalent to choosing in one step from their cartesian product:

do {x ← uniform xs ; y ← uniform ys ; return (x , y)} = uniform (cp xs ys)

where

cp :: [a ]→ [b ]→ [(a, b)]
cp xs ys = [(x , y) | x ← xs, y ← ys ]

We can prove this property by induction over xs, using equational reasoning
with the laws of MonadProb. For the base case of singleton lists, we have:

uniform (cp [x ] ys)
= [[ definition of cp ]]
uniform [(z , y) | z ← [x ], y ← ys ]

= [[ comprehensions: [f z | z ← [x ], p ] = [f x | p ] ]]
uniform [(x , y) | y ← ys ]

= [[ comprehensions: [f x | x ← xs ] = map f xs ]]
uniform (map (λy → (x , y)) ys)

= [[ naturality: uniform ◦map f = fmap f ◦ uniform ]]
do {y ← uniform ys ; return (x , y)}

= [[ left unit ]]
do {z ← return x ; y ← uniform ys ; return (z , y)}

= [[ definition of uniform ]]
do {z ← uniform [x ] ; y ← uniform ys ; return (z , y)}

and for the inductive step, assuming the result for xs, we have:

uniform (cp (x : xs) ys)
= [[ definition of cp ]]
uniform [(z , y) | z ← x : xs, y ← ys ]

= [[ comprehensions distribute over ++ ]]
uniform ([(z , y) | z ← [x ], y ← ys ] ++ [(z , y) | z ← xs, y ← ys ])

= [[ as above; definition of cp ]]
uniform (map (λy → (x , y)) ys ++ cp xs ys)

= [[ uniform distributes over ++; let n = length ys, l = length (cp xs ys) ]]
uniform (map (λy → (x , y)) ys) / n/n+l . uniform (cp xs ys)

= [[ let m = length xs, so l = m × n ]]
uniform (map (λy → (x , y)) ys) / 1/1+m . uniform (cp xs ys)

= [[ base case, inductive hypothesis ]]
do {z ← uniform [x ] ; y ← uniform ys ; return (z , y)} / 1/1+m .
do {z ← uniform xs ; y ← uniform ys ; return (z , y)}

= [[ composition distributes leftwards over / . ]]
do {z ← uniform [x ] / 1/1+m . uniform xs ; y ← uniform ys ; return (z , y)}

= [[ definition of uniform ]]
do {z ← uniform (x : xs) ; y ← uniform ys ; return (z , y)}

The second step uses the property

[f z ∨ z ← zs ++ zs ′, p ] = [f z ∨ z ← zs, p ] ++ [f z ∨ z ← zs ′, p ]

Yet again, simple equational reasoning suffices.

5 Combining algebraic theories

We have seen algebraic theories separately characterizing nondeterministic and
probabilistic choice. It is relatively straightforward to combine these two separate
algebraic theories into one integrated theory incorporating both nondeterminis-
tic and probabilistic choice. No new operations are required; the operations of
MonadAlt and MonadProb together suffice:

class (MonadAlt m,MonadProb m)⇒ MonadAltProb m

This Haskell type class declaration is complete; it has an empty collection of ad-
ditional methods, beyond those inherited from the superclasses MonadAlt and
MonadProb. Implicitly, the laws of MonadAlt and MonadProb are also inher-
ited; the only effort required is to consider the behaviour of interactions between
the methods of the two superclasses. We stipulate that probabilistic choice dis-
tributes over nondeterministic:

p / w . (q � r) = (p / w . q) � (p / w . r)

(This is not an uncontentious decision—some authors [55, 39] impose the oppo-
site distributivity, of nondeterministic choice over probabilistic; we discuss this
further in Sections 5.2 and 9.1.)

It turns out that there is a simple implementation of the combined interface,
as finite non-empty sets of distributions. Again, we approximate finite sets by
lists, for simplicity:

type Dists a = [Dist a ] -- nonempty lists

But the justification for this implementation is a little involved. The composition
as functors F G of two monads F ,G does not necessarily yield a monad: it is

straightforward to provide appropriate definitions of return and fmap, but not
always possible to define join (or, equivalently, >>=). However, it is a standard
result [2] that the composite F G does form a monad if there is a ‘distributive law
of G over F ’—that is, a natural transformation swap : G F → F G satisfying
certain coherence conditions. Given swap, it is also straightforward to define
join : F G F G → F F G G → F G ; that join satisfies the monad laws then
follows from the coherence conditions on swap.

In programming terms, we have to provide a distributive law of distributions
over lists

swap :: Dist [a ]→ [Dist a ] -- nonempty lists

satisfying the following four coherence conditions:

swap ◦ fmapD returnL = returnL

swap ◦ returnD = fmapL returnD

swap ◦ fmapD joinL = joinL ◦ fmapL swap ◦ swap
swap ◦ joinD = fmapL joinD ◦ swap ◦ fmapD swap

(where, to be explicit about typing, we have subscripted each use of return, fmap,
and join with L or D to indicate the list and distribution instances, respectively).
Then we can declare that the composite datatype Dists forms a monad, following
the standard construction [2]:

instance Monad Dists where
return x = return (return x )
fmap f p = fmap (fmap f ) p
join pp = fmap join (join (map swap pp))

A suitable definition of swap is as follows:

swap = foldr1 pick ◦map split where
split (xs,w) = [[(x ,w)] | x ← xs ]
pick xds yds = [xd ++ yd | xd ← xds, yd ← yds ]

(Here, foldr1 is a variant of foldr for non-empty lists, taking only a binary op-
erator and no starting value.) Informally, swap takes a distribution of nondeter-
ministic choices to a nondeterministic choice of distributions, multiplying out all
the possibilities; for example,

swap ([([1, 2], 1/3), ([3, 4], 2/3)]) = [[(1, 1/3), (3, 2/3)], [(1, 1/3), (4, 2/3)],
[(2, 1/3), (3, 2/3)], [(2, 1/3), (4, 2/3)]]

The composite monad Dists inherits MonadAlt and MonadProb functionality
straightforwardly from its two component parts:

instance MonadAlt Dists where
p � q = p ++ q

instance MonadProb Dists where
p / w . q = [xd / w . yd | xd ← p, yd ← q ]

It is therefore an instance of the integrated theory of nondeterministic and prob-
abilistic choice:

instance MonadAltProb Dists

Of course, we should check distributivity too; we return to this point in Sec-
tion 5.2 below.

5.1 Example: mixing choices

Analogous to the fair coin, here is a biased coin:

coin ′ :: MonadProb m ⇒ Prob → m Bool
coin ′

w = return True / w . return False

(we write the parameter w as a subscript) and an arbitrary nondeterministic
choice between booleans:

arb :: MonadAlt m ⇒ m Bool
arb = return True � return False

And here are two programs that each make an arbitrary choice and a probabilistic
choice and compare them, but do so in different orders [12, 31]:

arbcoin, coinarb :: MonadAltProb m ⇒ Prob → m Bool
arbcoin w = do {a ← arb ; c ← coin ′

w ; return (a c)}
coinarb w = do {c ← coin ′

w ; a ← arb ; return (a c)}

Intuitively, because the probabilistic choice happens ‘first’ in coinarb, the nonde-
terministic choice can depend on it; whereas in arbcoin, the probabilistic choice
happens ‘last’, so the nondeterministic choice cannot depend on it—and more-
over, the probabilistic choice cannot depend on the nondeterministic either, be-
cause it would not be probabilistic if it did so. We can capture this intuition
calculationally, using the equational theory of the two kinds of choice. On the
one hand, we have:

arbcoin w
= [[ definition of arbcoin ]]

do {a ← arb ; c ← coin ′
w ; return (a c)}

= [[ definition of arb ]]
do {a ← (return True � return False) ; c ← coin ′

w ; return (a c)}
= [[ composition distributes leftwards over � ]]

do {a ← return True ; c ← coin ′
w ; return (a c)} �

do {a ← return False ; c ← coin ′
w ; return (a c)}

= [[ left unit, booleans ]]

do {c ← coin ′
w ; return c} � do {c ← coin ′

w ; return (¬ c)}
= [[ right unit; definition of coin ′

w ]]
coin ′

w � coin ′
w

On the other hand,

coinarb w
= [[ definition of coinarb ]]

do {c ← coin ′
w ; a ← arb ; return (a c)}

= [[ definition of coin ′
w ]]

do {c ← (return True / w . return False) ; a ← arb ; return (a c)}
= [[ composition distributes leftwards over / . ]]

do {c ← return True ; a ← arb ; return (a c)} / w .
do {c ← return False ; a ← arb ; return (a c)}

= [[ left unit, booleans ]]
do {a ← arb ; return a } / w . do {a ← arb ; return (¬ a)}

= [[ right unit; definition of arb ]]
(return True � return False) / w . (return False � return True)

= [[ commutativity of � ]]
(return True � return False) / w . (return True � return False)

= [[ idempotence of / . ]]
return True � return False

= [[ definition of arb ]]
arb

That is, the nondeterminism in arbcoin can be resolved only by choosing the dis-
tribution provided by coin ′

w itself, or its opposite—the nondeterministic choice
happens first, and depending on whether True or False is chosen, the proba-
bilistic choice has chance either w or w of matching it. In particular, if w = 1/2,
then the nondeterministic choice cannot influence the final outcome. But in
coinarb, the probabilistic choice happens first, and the subsequent nondetermin-
istic choice has complete freedom to enforce any outcome.

5.2 Convex closure

At the start of Section 5, we said that collections of distributions form a model
of the combined theory MonadAltProb. In fact, this is not quite right: strictly
speaking, there is no distributive law of distributions over sets [49], so the com-
position of the two monads is not a monad. Indeed, distribution of / . over �
and idempotence of / . together imply a convexity property:

p � q
= [[ idempotence of / .; arbitrary w ]]

(p � q) / w . (p � q)
= [[ distributing / . over � ]]

(p / w . p) � (q / w . p) � (p / w . q) � (q / w . q)

That is, if any two distributions p and q are possible outcomes, then so is any
convex combination p / w . q of them. As a consequence, we should consider
equivalence of collections of distributions up to convex closure. In particular, for
coinarb we have:

coinarb v
= [[ calculation in previous section ]]
return True � return False

= [[ / . distributes over � , as above; arbitrary w ]]
(return True / w . return False) � (return False / w . return False) �
(return True / w . return True) � (return False / w . return True)

= [[ commutativity and idempotence of / .; definition of coin ′
w ]]

coin ′
w � return False � return True � coin ′

w

and so the possible outcomes of coinarb v include all convex combinations of the
four basic distributions coin ′

w , return False, return True, and coin w , which as
it happens encompasses all possible distributions of the booleans.

This convexity intuition is computationally reasonable, if one considers re-
peated executions of a computation such as coin ′

1/2 � coin ′
1/3. If the nondeter-

minism is always resolved in favour of the fair coin, the result will be heads half
the time; if the nondeterminism is always resolved in favour of the biased coin,
the result will be heads one third of the time. But if resolution of the nondeter-
minism alternates evenly between the two, the result will be heads five-twelfths
of the time. Ove repeated executions, any distribution between the two extremes
can be obtained by some long-term strategy for resolving the nondeterminism;
but no strategy will yield a distribution outside the two extremes.

One might wonder why one distributive law (of probabilistic choice over
nondeterministic) should hold, while the other (of nondeterministic choice over
probabilistic) need not. It turns out that the latter does not match intuitions
about behaviour; for example, it is straightforward to calculate as follows:

p / w . q
= [[ idempotence of � ]]

(p / w . q) � (p / w . q)
= [[ assuming that � distributes over / . ]]

((p � p) / w . (p � q)) / w . ((q � p) / w . (q � q))
= [[ idempotence and commutativity of � ]]

(p / w . (p � q)) / w . ((p � q) / w . q)
= [[ flattened choices ]]
〈p@w2, (p � q)@2w w , q@w2〉

= [[ rearranging and renesting choices ]]

(p / w2

/w2+w2 . q) / w2 + w2 . (p � q)

Informally, any straight probabilistic choice is inherently polluted with some
taint of nondeterministic choice too. For example, letting w = 1/2 and w = 1/3
respectively, we can conclude that

p / 1/2 . q = (p / 1/2 . q) / 1/2 . (p � q)
p / 1/3 . q = (p / 1/5 . q) / 5/9 . (p � q)

This seems quite unfortunate, and so we do not require that nondeterministic
choice distributes over probabilistic.

6 Monty Hall

As an extended example, we turn to the so-called Monty Hall Problem [45], which
famously caused a controversy following its discussion in Marilyn vos Savant’s
column in Parade magazine in 1990 [50]. Vos Savant described the problem as
follows, quoting a letter from a reader, Craig F. Whitaker:

Suppose you’re on a game show, and you’re given the choice of three doors:
Behind one door is a car; behind the others, goats. You pick a door, say
No. 1, and the host, who knows what’s behind the doors, opens another
door, say No. 3, which has a goat. He then says to you, “Do you want to
pick door No. 2?” Is it to your advantage to switch your choice?

Implicit in the above statement, the car is equally likely to be behind each of the
three doors, the car is the prize and the goats are booby prizes, the host always
opens a door, which always differs from the one you pick and always reveals a
goat, and you always get the option to switch.

We might model this as follows. There are three doors:

data Door = A | B | C
doors :: [Door ]
doors = [A,B ,C ]

First, Monty hides the car behind one of the doors, chosen uniformly at random:

hide :: MonadProb m ⇒ m Door
hide = uniform doors

Second, you pick one of the doors, also uniformly at random:

pick :: MonadProb m ⇒ m Door
pick = uniform doors

Third, Monty teases you by opening one of the doors—not the one that hides
the car, nor the one you picked—to reveal a goat, choosing randomly among the
one or two doors available to him:

tease :: MonadProb m ⇒ Door → Door → m Door
tease h p = uniform (doors \\ [h, p ])

(Here, the expression xs \\ ys denotes the list of those elements of xs absent
from ys.) Fourth, Monty offers you the choice between two strategies—either to
switch to the door that is neither your original choice nor the opened one:

switch :: MonadProb m ⇒ Door → Door → m Door
switch p t = return (head (doors \\ [p, t ]))

or to stick with your original choice:

stick :: MonadProb m ⇒ Door → Door → m Door
stick p t = return p

In either case, you know p and t—but not h, of course.
Here is the whole game, parametrized by your strategy, returning whether

you win the car:

monty :: MonadProb m ⇒ (Door → Door → m Door)→ m Bool
monty strategy

= do {h ← hide ; -- Monty hides the car behind door h
p ← pick ; -- you pick door p
t ← tease h p ; -- Monty teases you with door t (6= h, p)
s ← strategy p t ; -- you choose, based on p and t but not h
return (s h) -- you win iff your choice s equals h
}

We will show below that the switching strategy is twice as good as the sticking
strategy:

monty switch = coin ′
2/3

monty stick = coin ′
1/3

The key is the fact that separate uniform choices are independent:

do {h ← hide ; p ← pick ; return (h, p)}
= [[ definitions of hide and pick ]]

do {h ← uniform doors ; p ← uniform doors ; return (h, p)}
= [[ independent choices ]]
uniform (cp doors doors)

and so we have

monty strategy = do
(h, p)← uniform (cp doors doors)
t ← tease h p
s ← strategy p t
return (s h)

Clearly, the doors h and p independently chosen at random will match one third
of the time:

do {(h, p)← uniform (cp doors doors) ; return (p h)}
= [[ fmap and do notation ]]

do {b ← fmap (uncurry ( )) (uniform (cp doors doors)) ; return b}
= [[ right unit ]]
fmap (uncurry ( )) (uniform (cp doors doors))

= [[ naturality of uniform ]]
uniform (map (uncurry ( )) (cp doors doors))

= [[ definitions of doors, cp, ]]
uniform [True,False,False,False,True,False,False,False,True ]

= [[ simplifying: three Trues, six Falses ]]
uniform [True,False,False ]

= [[ definitions of uniform, coin ′
w ]]

coin ′
1/3

Therefore we calculate:

monty stick
= [[ definition of monty , independent uniform choices ]]

do {(h, p)← uniform (cp doors doors) ;
t ← tease h p ; s ← stick p t ; return (s h)}

= [[ definition of stick ]]
do {(h, p)← uniform (cp doors doors) ;

t ← tease h p ; s ← return p ; return (s h)}
= [[ left unit ]]

do {(h, p)← uniform (cp doors doors) ; t ← tease h p ; return (p h)}
= [[ t unused, and uniform side-effect-free, so tease can be eliminated ]]

do {(h, p)← uniform (cp doors doors) ; return (p h)}
= [[ matching choices, as above ]]
coin ′

1/3

and

monty switch
= [[ definition of monty , independent uniform choices ]]

do {(h, p)← uniform (cp doors doors) ; t ← tease h p ;
s ← switch p t ; return (s h)}

= [[ definition of switch ]]
do {(h, p)← uniform (cp doors doors) ; t ← tease h p ;

s ← return (head (doors \\ [p, t ])) ; return (s h)}
= [[ left unit ]]

do {(h, p)← uniform (cp doors doors) ; t ← tease h p ;
return (h head (doors \\ [p, t ]))}

= [[ case analysis on h = p—see below ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then return False else return True }
= [[ booleans ]]

do {(h, p)← uniform (cp doors doors) ; return (h 6= p)}
= [[ analogously, mismatching choices ]]
coin ′

2/3

Now for the two branches of the case analysis. For the case h = p, we have:

do {t ← tease h p ; return (h head (doors \\ [p, t ]))}
= [[ using assumption h p ]]

do {t ← tease h p ; return (h head (doors \\ [h, t ]))}
= [[ h is not in doors \\ [h, t ] ]]

do {t ← tease h p ; return False }
= [[ t unused, and uniform side-effect-free ]]
return False

And for the case h 6= p, we have:

do {t ← tease h p ; return (h head (doors \\ [p, t ]))}
= [[ definition of tease ]]

do {t ← uniform (doors \\ [h, p ]) ; return (h head (doors \\ [p, t ]))}
= [[ h 6= p, so doors \\ [h, p ] is a singleton; uniform [a ] = return a ]]

do {t ← return (head (doors \\ [h, p ])) ; return (h head (doors \\ [p, t ]))}
= [[ left unit ]]

do {let t = head (doors \\ [h, p ]) ; return (h head (doors \\ [p, t ]))}
= [[ h 6= p, and t 6= h, p; so t , h, p distinct ]]

do {let t = head (doors \\ [h, p ]) ; return (h h)}
= [[ t unused ]]
return True

So when you and Monty make uniform probabilistic choices according to the
rules of the game, switching wins two thirds of the time and sticking only one
third.

6.1 Nondeterministic Monty

Perhaps a more faithful model of the Monty Hall problem is to allow Monty to
make nondeterministic rather than probabilistic choices [31]—nobody said that
Monty has to play fair. That is, Monty’s two moves in the game, hiding the
car and teasing you, involve a nondeterministic rather than probabilistic choice
among the available alternatives:

hiden :: MonadAlt m ⇒ m Door
hiden = arbitrary doors

teasen :: MonadAlt m ⇒ Door → Door → m Door
teasen h p = arbitrary (doors \\ [h, p ])

where

arbitrary :: MonadAlt m ⇒ [a ]→ m a
arbitrary = foldr1 (�) ◦map return

Then we define the game just as before, but with Monty behaving nondetermin-
istically:



Abstraction and Reasoning for Probabilistic Systems 27

montyn :: MonadAltProb m ⇒ (Door → Door → m Door)→ m Bool
montyn strategy = do {h ← hiden ; p ← pick ; t ← teasen h p ;

s ← strategy p t ; return (s h)}

As it happens, making this change has no effect on the outcome. The first
two choices—Monty’s choice of where to hide the car, and your initial choice
of door—can still be combined, because composition distributes leftwards over
nondeterministic choice:

do {h ← hiden ; p ← pick ; return (h, p)}
= [[ let k h = do {p ← pick ; return (h, p)} ]]

do {h ← hiden ; k h }
= [[ definition of hiden, arbitrary ]]

do {h ← (return A � return B � return C ) ; k h }
= [[ composition distributes leftwards over � ]]

do {h ← return A ; k h } � do {h ← return B ; k h } �
do {h ← return C ; k h }

= [[ left unit ]]
k A � k B � k C

= [[ definition of k ]]
do {p ← pick ; return (A, p)} � do {p ← pick ; return (B , p)} �
do {p ← pick ; return (C , p)}

The remainder of the reasoning proceeds just as before. It is still the case that
doors h and p will match one third of the time, even though h is now chosen
nondeterministically rather than probabilistically. For brevity, let

try d = do {p ← pick ; return (d , p)}

Then we have

fmap (uncurry ( )) (try d)
= [[ definition of try ; fmap and do notation ]]

do {p ← pick ; return (d p)}
= [[ definition of pick ]]

do {p ← 〈return A@1/3, return B@1/3, return C@1/3〉 ; return (d p)}
= [[ composition distributes leftwards over / .; right unit ]]
〈return (d A)@1/3, return (d B)@1/3, return (d C )@1/3〉

= [[ d :: Door , and so d is one of A,B ,C ]]
〈return True@1/3, return False@1/3, return False@1/3〉

= [[ definition of coin ′
w ]]

coin ′
1/3

and therefore

do {h ← hiden ; p ← pick ; return (h p)}
= [[ fmap and do notation ]]

fmap (uncurry ( )) (do {h ← hiden ; p ← pick ; return (h, p)})
= [[ combining first two choices, as above ]]
fmap (uncurry ( )) (try A � try B � try C )

= [[ naturality: fmap f (p � q) = fmap f p � fmap f q ]]
fmap (uncurry ( )) (try A) � fmap (uncurry ( )) (try B) �
fmap (uncurry ( )) (try C )

= [[ matching choices, above ]]
coin ′

1/3 � coin ′
1/3 � coin ′

1/3

= [[ idempotence of � ]]
coin ′

1/3

and the conclusion is still that

montyn switch = coin ′
2/3

montyn stick = coin ′
1/3

Combining two classes of effect—here, probability and nondeterminism—did not
make the reasoning any more difficult that in was with a single such class.

7 Failure is an option

Computations that may fail, and whose failures can be handled, are characterized
by two operations for throwing and catching exceptions:

class Monad m ⇒ MonadExcept m where
throw :: m a
catch :: m a → m a → m a

For simplicity, we suppose that there is only a single exception, just as we as-
sumed a single updatable location for stateful computations; the model is easily
extended to cover multiple exceptions. The intuition is that throw is the com-
putation that immediately fails, and that p ‘catch‘ q represents the computation
that behaves like p, except if this fails, in which case it continues as the exception
handler q . (In Haskell, backquotes turn a prefix function into an infix operator;
so p ‘catch‘ q = catch p q .) We stipulate that throw is a left zero of composition,
so that a failure discards the subsequent part of a computation:

do {x ← throw ; k x } = throw

We do not stipulate that throw is a right zero of composition; that would require
that a failure also discards the preceding part of the computation, and so that
any effects of that part would have to be rolled back—quite a strong condition.
Neither do we stipulate that composition distributes over catch; in general,

do {x ← (p ‘catch‘ q) ; k x } 6= do {x ← p ; k x } ‘catch‘ do {x ← q ; k x }

because the right-hand side brings exceptions raised by the first k under the
influence of the handler q , whereas exceptions of k on the left are not handled.

We do stipulate that return is a left zero of catch, so that pure computations
never fail:

return x ‘catch‘ p = return x

Finally, throw and catch form a monoid:

p ‘catch‘ throw = p
throw ‘catch‘ p = p
p ‘catch‘ (q ‘catch‘ r) = (p ‘catch‘ q) ‘catch‘ r

That is: an exception handler that immediately propagates the exception has no
effect; failures are indeed caught and handled; and exception handlers can be
chained in sequence, with control passing along the chain as failures happen.

One obvious implementation of exceptions is via lifting:

data Maybe a = Just a | Nothing

Values are lifted into pure computations via Just and passed along by composi-
tion, whereas Nothing forms a left zero of composition:

instance Monad Maybe where
return x = Just x
Just x >>= k = k x
Nothing >>= k = Nothing

Of course, Nothing represents failure; Just and Nothing partition computations
into entirely pure ones and entirely failing ones, which form a left zero and a left
unit of catch, respectively.

instance MonadExcept Maybe where
throw = Nothing
Just x ‘catch‘ q = Just x
Nothing ‘catch‘ q = q

The names Maybe, Just , and Nothing were coined by Spivey [46], and he gives
a number of examples of equational reasoning about term rewriting operations
that might fail. (In ML, the datatype analogous to Maybe a is written ′a option.)

7.1 Combining probability with exceptions

Just as we did for nondeterministic and probabilistic choice, we can quite easily
combine the theories of probability and exceptions. We simply combine the two
interfaces, adding no new operations:

class (MonadExcept m,MonadProb m)⇒ MonadProbExcept m

The laws of exceptions and of probability are inherited; the only effort required
is to consider the interaction between the two theories. In this case, we have one

additional distributivity law, which intuitively states that making a probabilistic
choice cannot of itself cause a failure, so catch distributes over it:

(p / w . q) ‘catch‘ r = (p ‘catch‘ r) / w . (q ‘catch‘ r)

The same representation works as for plain probability distributions, but now
we allow the weights to sum to less than one, and the list of weighted elements to
be empty—these are sometimes called subdistributions [31] or evaluations [20]:

weight :: Dist a → Prob
weight p = sum [w | (x ,w)← p ]

instance MonadExcept Dist where
throw = [ ]
p ‘catch‘ q = p ++ scale (1−weight p) (q)

instance MonadProbExcept Dist

Operationally, the exceptions are represented by the ‘missing’ bits of the distri-
bution; for example, the subdistribution [(True, 1/2), (False, 1/4)] has weight 3/4,
and represents a computation that fails the remaining 1/4 of the time. The cor-
rectness of this implementation depends on the congruences we have imposed on
distributions, specifically to ignore reordering and zero-weighted elements, and
to coalesce duplicates.

For example, here is an attempt to simulate the biased coin ′
2/3 using two

fair coin tosses, motivated by Knuth and Yao’s trick [23] for simulating a fair
die with three coins:

coins23 :: MonadProbExcept m ⇒ m Bool
coins23 = return True / 1/2 . (return False / 1/2 . throw)

We might illustrate the process as follows:

•
head
yy

tail

##
True •

head

||
tail

##
False throw

This does indeed yield True exactly twice as often as False. However, a quarter
of the time it will fail to yield any result at all, and throw an exception instead;
so the attempt was not entirely successful. We will pick up this example again
in Section 8 below.

7.2 Forgetful Monty

Let us return to the purely probabilistic version of the Monty Hall game, but this
time suppose that Monty is becoming increasingly forgetful as he turns sixty—
he can never remember where he has hidden the car [45, Chapter 3]. Therefore,

when it comes to teasing you, he uniformly at random opens one of the two
doors different from the door you picked. Of course he might accidentally reveal
the car in doing so; we treat this as a failure in the protocol:

teasef :: MonadProbExcept m ⇒ Door → Door → m Door
teasef h p = do {t ← uniform (doors \\ [p ]) ; if t h then throw else return t }
montyf :: MonadProbExcept m ⇒ (Door → Door → m Door)→ m Bool

montyf strategy = do {h ← hide ;

p ← pick ;
t ← teasef h p ;
s ← strategy p t ;
return (s h)}

Investigating teasef in the case that h = p, we have:

teasef h p
= [[ definition of teasef ]]

do {t ← uniform (doors \\ [p ]) ; if t h then throw else return t }
= [[ h = p, so h is not in doors \\ [p ] and hence t 6= h ]]

do {t ← uniform (doors \\ [p ]) ; return t }
= [[ right unit ]]
uniform (doors \\ [p ])

In the case that h 6= p, let d = head (doors \\ [h, p ]), so that h, p, d are all
distinct; then we have:

teasef h p
= [[ definition of teasef ]]

do {t ← uniform (doors \\ [p ]) ; if t h then throw else return t }
= [[ by assumption, doors \\ [p ] = [h, d ] ]]

do {t ← uniform [h, d ] ; if t h then throw else return t }
= [[ definition of uniform ]]

do {t ← return h / 1/2 . return d ; if t h then throw else return t }
= [[ composition distributes leftwards over / . ]]

do {t ← return h ; if t h then throw else return t } / 1/2 .
do {t ← return d ; if t h then throw else return t }

= [[ left unit ]]
(if h h then throw else return h) / 1/2 . (if d h then throw else return d)

= [[ by assumption, d 6= h; conditionals ]]
throw / 1/2 . return d

and so

teasef h p = if h p then uniform (doors \\ [p ])
else (throw / 1/2 . return (head (doors \\ [h, p ])))

Clearly, in the ‘else’ case, teasef is no longer necessarily side-effect-free—even
if its result is not used, it cannot be discarded, because it might fail—and so

the calculations for the purely probabilistic version of the game do not apply.
Instead, we have:

montyf stick

= [[ definition of montyf ]]

do {(h, p)← uniform (cp doors doors) ;
t ← teasef h p ; s ← stick p t ; return (s h)}

= [[ definition of stick ; left unit ]]
do {(h, p)← uniform (cp doors doors) ; t ← teasef h p ; return (p h)}

= [[ case analysis in teasef , as above; let d = head (doors \\ [h, p ]) ]]
do {(h, p)← uniform (cp doors doors) ;

t ← if h p then uniform (doors \\ [p ]) else (throw / 1/2 . return d) ;
return (p h)}

= [[ conditionals ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then do {t ← uniform (doors \\ [p ]) ; return (p h)}
else do {t ← throw / 1/2 . return d ; return (p h)}}

= [[ first t is unused and uniform is side-effect-free; p h = True ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then return True
else do {t ← throw / 1/2 . return d ; return (p h)}}

= [[ composition distributes over / . ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then return True
else do {t ← throw ; return False } / 1/2 .

do {t ← return d ; return False }}
= [[ throw is a left zero of composition, and return a left unit ]]

do {(h, p)← uniform (cp doors doors) ;
if h p then return True else throw / 1/2 . return False }

= [[ matching choices, as in Section 6 ]]
do {b ← coin ′

1/3 ; if b then return True else throw / 1/2 . return False }
= [[ composition distributes over / . ]]

do {b ← return True ;
if b then return True else throw / 1/2 . return False } / 1/3 .

do {b ← return False ;
if b then return True else throw / 1/2 . return False }

= [[ left unit; conditionals ]]
return True / 1/3 . (throw / 1/2 . return False)

= [[ flattening choices ]]
〈True@1/3, throw@1/3,False@1/3〉

On the other hand:

montyf switch

= [[ definition of montyf ]]

do {(h, p)← uniform (cp doors doors) ; t ← teasef h p ;

s ← switch p t ; return (s h)}
= [[ definition of switch ]]

do {(h, p)← uniform (cp doors doors) ; t ← teasef h p ;
s ← return (head (doors \\ [p, t ])) ; return (s h)}

= [[ left unit ]]
do {(h, p)← uniform (cp doors doors) ; t ← teasef h p ;

return (h head (doors \\ [p, t ]))}
= [[ case analysis in teasef , as above; let d = head (doors \\ [h, p ]) ]]

do {(h, p)← uniform (cp doors doors) ;
t ← if h p then uniform (doors \\ [p ]) else throw / 1/2 . return d ;
return (h head (doors \\ [p, t ]))}

= [[ composition distributes leftwards over conditional ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then do {t ← uniform (doors \\ [p ]) ;
return (h head (doors \\ [p, t ]))}

else do {t ← throw / 1/2 . return d ;
return (h head (doors \\ [p, t ]))}}

= [[ in then branch, h = p, so h is not in doors \\ [p, t ] ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then do {t ← uniform (doors \\ [p ]) ; return False }
else do {t ← throw / 1/2 . return d ;

return (h head (doors \\ [p, t ]))}}
= [[ composition distributes over / . ]]

do {(h, p)← uniform (cp doors doors) ;
if h p then do {t ← uniform (doors \\ [p ]) ; return False }

else do {t ← throw ; return (h head (doors \\ [p, t ]))} / 1/2 .
do {t ← return d ; return (h head (doors \\ [p, t ]))}}

= [[ in then branch, t is unused and uniform is side-effect-free ]]
do {(h, p)← uniform (cp doors doors) ;

if h p then return False
else do {t ← throw ; return (h head (doors \\ [p, t ]))} / 1/2 .

do {t ← return d ; return (h head (doors \\ [p, t ]))}}
= [[ throw is left zero, return is left unit ]]

do {(h, p)← uniform (cp doors doors) ;
if h p then return False

else throw / 1/2 . (return (h head (doors \\ [p, d ])))}
= [[ h, p, d are distinct, so head (doors \\ [p, d ]) = h ]]

do {(h, p)← uniform (cp doors doors) ;
if h p then return False else throw / 1/2 . (return True)}

= [[ matching choices ]]
do {b ← coin ′

1/3 ; if b then return False else throw / 1/2 . return True }
= [[ composition distributes over / . ]]

do {b ← return True ;
if b then return False else throw / 1/2 . return True } / 1/3 .

do {b ← return False ;

if b then return False else throw / 1/2 . return True }
= [[ left unit; conditionals ]]
return False / 1/3 . (throw / 1/2 . return True)

= [[ flattening choices ]]
〈False@1/3, throw@1/3,True@1/3〉

So, somewhat surprisingly, both the sticking and switching strategies are equiva-
lent in the face of Monty’s forgetfulness: with either one, you have equal chances
of winning, losing, or of the game being aborted.

montyf stick = montyf switch = 〈True@1/3,False@1/3, throw@1/3〉

7.3 Conditional probability

We have so far presented the combination of probability and exceptions as mod-
elling computations that make probabilistic choices, but that might fail. An
alternative reading is in terms of conditional probability. In probability theory,
the conditional probability P (A|B) is the probability of event A occurring, given
that event B is known to have occurred; it is defined to be the probability of
both events A and B occurring, divided by the probability of B alone:

P (A |B) =
P (A ∧B)

P (B)

Operationally, a subdistribution p with weight w > 0 represents the distribution
of outcomes drawn from the normalized distribution scale (1/w ) p, conditioned
by the outcomes outside the support of p not occurring.

normalize :: Dist a → Dist a
normalize p = scale (1 / weight p) (p)

For example, the subdistribution

coins23 = [(True, 1/2), (False, 1/4)]

from Section 7.1 has weight 3/4, and so represents the distribution of outcomes

[(True, 1/2 ÷ 3/4 = 2/3), (False, 1/4 ÷ 3/4 = 1/3)] = coin ′
2/3

given that one does not toss two tails—so the attempt to simulate the biased
coin ′

2/3 using two fair coin tosses was not so far off after all. Similarly, one
could say that playing against forgetful Monty using either the switching or
the sticking strategy yields a 50–50 chance of winning, assuming that Monty
successfully bluffs his way through his amnesia.

As a more extended example, consider the canonical ‘wet grass’ Bayesian
reasoning problem [21]. Suppose that with probability 3/10 it is raining; and
when it rains, with probability 9/10 it does so heavily enough to make the grass

wet. Also, there is a lawn sprinker, operating with probability 1/2; when this
operates, with probability 8/10 it has high enough water pressure to make the
grass wet. Finally, with probability 1/10 the grass is wet for some other unknown
reason.

rain :: MonadProb m ⇒ m Bool
rain = coin ′

3/10

sprinkler :: MonadProb m ⇒ m Bool
sprinkler = coin ′

1/2

grassWet :: MonadProb m ⇒ Bool → Bool → m Bool
grassWet r s = do {x ← coin ′

9/10 ; y ← coin ′
8/10 ; z ← coin ′

1/10 ;
return ((x ∧ r) ∨ (y ∧ s) ∨ z )}

What is the probability that it is raining, given that the grass is observed to be
wet?

experiment :: MonadProbExcept m ⇒ m Bool
experiment = do {r ← rain ; s ← sprinkler ; g ← grassWet r s ;

if g then return r else throw }

We simply return whether it is raining, conditioned on whether the grass is wet:

normalize experiment = [(False, 1610/3029), (True, 1419/3029)]

—that is, it is raining with probability 1419/3029 ' 0.47.

8 Recursion

It is very tempting to write recursive programs in the style we have shown
above. For example, here is a simple Markov model of the coins23 attempt in
Section 7.1:

•
head
yy

tail

##
True •

head

||

tail
ww

False

This depicts a process that recurses instead of failing. We can represent it sym-
bolically with a recursive definition:

thirds :: MonadProbExcept m ⇒ m Bool
thirds = coins23 ‘catch‘ thirds

In fact, we can inline the exception and its handler, on account of the various
laws of / .: for any p,

coins23 ‘catch‘ p
= [[ definition of coins23 ]]

(return True / 1/2 . (return False / 1/2 . throw)) ‘catch‘ p
= [[ catch distributes over / ., twice ]]

(return True ‘catch‘ thirds) / 1/2 .
((return False ‘catch‘ thirds) / 1/2 . (throw ‘catch‘ p))

= [[ return is left zero of catch, and throw is left unit ]]
return True / 1/2 . (return False / 1/2 . p)

So we could have defined instead

thirds :: MonadProb m ⇒ m Bool
thirds = return True / 1/2 . (return False / 1/2 . thirds)

Note that this definition can be given a more specific type, because it no longer
depends on the ability to fail.

8.1 Recursive definitions

But what might the semantics of such a recursive definition be? Up until now,
we have implicitly assumed a setting of sets and total functions, in which there is
no guarantee that an arbitrary recursive definition has a canonical solution. And
indeed, with the implementation of MonadProb in terms of finite probability-
weighted lists from Section 4, the recursive equation defining thirds has no so-
lution.

The usual response to this problem is to switch the setting from total func-
tions between sets to continuous functions between complete partial orders; then
all recursive definitions have a least solution. In this case, the least solution to
the recursive definition of thirds above is the infinite probability-weighted list

[(True, 1/2), (False, 1/4), (True, 1/8), (False, 1/16), (True, 1/32), (False, 1/64), ...]

One might see this as a reasonable representation of coin ′
2/3: the weights of fi-

nite initial segments of the list converge to one; and two thirds of the weight is
associated with True, one third with False. Moreover, if one samples from this
distribution in the obvious way, with probability 1 one obtains an appropriate
result—only in the measure-zero situation in which one tries to sample the dis-
tribution at precisely 1.0 does the computation not terminate. In fact, one can
show that coin ′

2/3 is a solution to the same recursive equation as thirds:

return True / 1/2 . (return False / 1/2 . coin
′
2/3)

= [[ definition of coin ′
w ]]

return True / 1/2 . (return False / 1/2 . (return True / 2/3 . return False))
= [[ flattening choices ]]
〈return True@1/2, return False@1/4, return True@1/6, return False@1/12〉

= [[ combining collisions ]]
〈return True@2/3, return False@1/3〉

= [[ definition of coin ′
w ]]

coin ′
2/3

However, the nice behaviour in this case is a happy accident of the rather
special form of the recursive definition. Had we written the ostensibly equivalent
definition

thirds = (thirds / 1/2 . return False) / 1/2 . return True

instead, the least solution would be ⊥, the least-defined element in the infor-
mation ordering, because the definition of / . on weighted lists is strict in its
left argument. In fact, the codatatype of possibly-infinite, possibly-partial lists
fails to satisfy all the necessary axioms: choice is not quasi-commutative when
it involves undefined arguments, because lists are left-biased.

8.2 A free monad for choice

As we have seen, one obvious implementation of MonadProb uses probability-
weighted lists of values; thus, coin is represented as [(True, 1/2), (False, 1/2)].
However, an arguably more natural representation is in terms of the free monad—
‘more natural’ in the sense that it arises directly from the signature of the / .
operation:

data DistT a = Return a | Choice Prob (DistT a) (DistT a)

This being a free monad, return and bind have simple definitions arising from
substitution:

instance Monad DistT where
return x = Return x
Return x >>= k = k x
Choice w p q >>= k = Choice w (p >>= k) (q >>= k)

and by design, / . is trivial to implement:

instance MonadProb DistT where
p / w . q = Choice w p q

Again, because this is the free monad, the monad laws necessarily hold—the left
unit law holds directly by construction, and the right unit and associativity laws
are easily shown by induction over the structure of the left-hand argument. This
is not quite the initial model of the MonadProb specification, though, because the
remaining identity, idempotence, quasi-commutativity, and quasi-associativity
laws of / . do not hold. Indeed, as we have already argued, the initial model
of the specification consists of finite mappings from elements to probabilities,
collected from the choice tree in the obvious way:

collectT :: Eq a ⇒ DistT a → (a → Prob)
collectT (Return x ) y = if x y then 1 else 0
collectT (Choice w p q) y = w × collectT p y + w × collectT q y

and equivalences on DistT ought again to be taken modulo permutations, zero-
weighted elements, and repeated elements (whose weights should be added).
Nevertheless, the datatype DistT itself provides another convenient approxima-
tion to the initial model.

In order to guarantee solutions to arbitrary recursive definitions, we still
have to accept the CPO setting rather than SET; so really, we mean to take the
codatatype interpretation of DistT , including partial and infinite values as well
as finite ones. (Which means that the inductive proofs of the monad laws need
to be strengthened to cover these cases.)

The benefit we gain from this extra complication is that the representation is
now symmetric, and so the laws of choice hold once more. Recursive definitions
like those of thirds above give rise to regular choice trees—possibly infinite trees,
but with only finitely many different subtrees. Moreover, / . is non-strict in both
arguments, so the semantics of recursive definitions is much less sensitive to the
precise form of the equation.

But of course, not all recursive equations give productive solutions. Clearly
anything is a solution of the equation p = p, or, thanks to the monad laws,
of p = (p >>= return). Even some equations that completely determine proper
choice trees do not define productive sampling behaviour; for example,

p = p / 1 . return True

does not, and even when restricting attention to weights strictly between 0 and
1, one can try

p = p / 1/2 . p

We believe that for a recursive definition to define a choice tree with a productive
interpretation as a sampling function, it is sufficient for recursive occurrences of
the variable being defined to be guarded by a / ., and for each such / . to
devote positive weight to a non-recursive subcomputation. But it seems very
unsatisfactory to have to consider specific implementations such as DistT of the
MonadProb specification at all; in the rest of the paper, we have managed to
conduct all our reasoning in terms of the algebraic theory rather than one of its
models.

9 Conclusions

We have presented an approach to reasoning about effectful functional programs,
based on algebraic theories. We have focussed in this paper on the effects of
nondeterminism and probabilism and their combination, in honour of Carroll
Morgan, but the approach works for any class of effects; our earlier paper [10] also

discusses counting, generating fresh names, mutable state, and exceptions. The
problem of reasoning about effectful programs was an open one in the functional
programming community before this point; for example, our work was inspired
by a paper concerning a relabelling operation on trees [18], which resorted to
unrolling the obvious stateful program into a pure state-transforming function
in order to conduct proofs.

One strength of functional programming is the support it provides for rea-
soning about programs: the techniques of simple high-school algebra suffice, and
one can reason directly in the language of programs, rather than having to ex-
tract verification conditions from the programs and then reason indirectly in
predicate calculus. In that respect, functional programming has a similar moti-
vation to Hoare’s ‘programs are predicates’ work [16], Hehner’s predicative pro-
gramming [13], and Morgan’s refinement calculus [37]—namely, to avoid where
possible the distinction between syntax and semantics, and to remove the layer
of interpretation that translates from the former to the latter.

The other main strength of functional programming is the tools it provides for
abstraction: for defining ‘embedded domain-specific languages’ in terms of exist-
ing language constructs such as algebraic datatypes, higher-order functions, and
lazy evaluation, rather than having to step outside the existing language to de-
fine a new one. In that respect, functional programming goes beyond predicative
programming and refinement calculus, which are not expressive enough to sup-
port such extension. (Of course, embedded DSLs have their limits. Sometimes a
desired feature cannot be defined conveniently, if at all, in terms of existing con-
structs; then the best approach is to relent and to define a completely different
language after all.)

9.1 Nondeterminism and probability in program calculi

Dijkstra [6] argued forcefully for the centrality of nondeterminism in program-
ming calculi, particularly in order to support underspecification and program
development by stepwise refinement. That impetus led to the development of
the refinement calculus [37, 1] for imperative programming. However, although
functional programming is an excellent setting in which to calculate with pro-
grams, it does not support refinement well—for that, one has to make the step
from functions to relations [3]. Nevertheless, functional programming is very
convenient for manipulating collections of elements, and collection types such
as lists and sets form monads; so collection-oriented programming does fit the
functional view well [51, 4].

Lawvere [27, 11] pointed out that probability distributions form a monad
too; this has led to a slow but steady stream of functional programming ap-
proaches to probabilistic computation [20, 44, 9, 21]. Independently, Kozen [24]
presented a semantics for while-programs with probabilistic choice; but it was a
longstanding challenge to integrate this semantics with nondeterministic choice.
There was a flurry of work in the early 1990s addressing this issue within the
process algebra setting [55, 29, 39]. He et al. [12] used Jones’ probabilistic pow-
erdomain construction to provide a semantics for a guarded command language

with both probabilistic and nondeterministic choice; in fact, they defined two
semantics—one like ours, in which, operationally speaking, demonic nondeter-
minism is resolved at run-time whenever a nondeterministic choice is executed,
and another is which nondeterminism is resolved at compile-time, but which sac-
rifices idempotence of conditional. The first of He et al.’s semantics is the basis of
the ‘demonic/probabilistic’ approach taken by Morgan [31]. Varacca [49], citing a
personal correspondence with Gordon Plotkin, shows that although the compo-
sition of the powerset and probability distribution monads do not directly form
a monad, this is fixed by taking the convex closure—giving rise to essentially
the model we have in Section 5.2.

The combination of nondeterminism and probability and the selection of dis-
tributivity properties that we have presented here are not new; they are fairly
well established in work on program calculi [12, 31, 32, 34]. Curiously, however,
not all authors settle on the same distributivity properties; some [55, 39] have
nondeterministic choice distributing over probabilistic, the opposite of the ap-
proach we take. Choosing this law sacrifices the intuitively reasonable arbcoin
example; by the same line of reasoning as in Section 5.2, one can show under
this alternative distributivity regime that, for arbitrary w ,

arbcoin v = (arb / w . return False) / w . (return True / w . arb)

That is, with probability w2+w2 (independent of v !) the outcome is an arbitrary
choice, and otherwise it is determined probabilistically. Worse, having nondeter-
ministic choice distribute over probabilistic is inconsistent with idempotence—by
similar reasoning again, using idempotence and commutativity of � and of / .,
and distributivity of � over / ., one can show that

coin = coin / 1/2 . arb

which seems a most unwelcome consequence: even a fair coin can be subverted.
One might argue (as Mislove [35] does) that idempotence is a fundamental prop-
erty of nondeterminism, and that this consequence for coin is untenable, and that
therefore the alternative distributivity property should be avoided.

Similarly, Deng et al. [5] show that taking the perspective of testing equiva-
lences on a probabilistic version of CSP—that is, attempting to distinguish two
CSP processes by exhibiting different outcomes when they are run in parallel
with a common test process—eliminates many of the otherwise reasonable equiv-
alences; in particular, they give counterexamples to either distributivity property
between nondeterministic (‘internal’) and probabilistic choice; however, distribu-
tivity of CSP’s ‘external’ choice over probabilistic choice does still survive.

However, we emphasize that our approach is agnostic as to the particular
axiomatization. It is perfectly possible to impose distributivity of nondetermin-
istic over probabilistic choice, obtaining the consequences for arbcoin and coin
above; or to impose no distributivity law at all, in which case there are sim-
ply fewer program equivalences. The approach still works; whether it faithfully
models intuition or reality is a separate question.



Abstraction and Reasoning for Probabilistic Systems 41

9.2 Beyond finite support

The approach we have presented here really only tells the story for finitely sup-
ported probability distributions. By exploiting recursive definitions, one might
hope to be able to build distributions with infinite support; for example, with

naturals :: MonadProb m ⇒ Integer → m Integer
naturals n = return n / 1/2 . naturals (n + 1)

one might hope that naturals 0 returns result i with probability 1/2i+1, possibly
returning any of the naturals. This works in a lazy language like Haskell, provided
that the definition of / . is non-strict in its right argument; for example, for an
implementation based on possibly infinite weighted lists as in Section 8.1, it
yields

[(0, 1/2), (1, 1/4), (2, 1/8), (3, 1/16), (4, 1/32), (5, 1/64), ...]

as expected. We are optimistic that the ‘free monad’ technique from Section 8.2
might be extended to give a more disciplined explanation of such cases. In re-
lated work [41], we have been using free monads to model a generic framework
for tracing execution. In the case of nonterminating computations, one will in
general get an infinite trace; to account for that case, one needs a coalgebraic
rather than an algebraic reading of the tracing datatype. Perhaps this can all be
conveniently captured in a ‘strong functional programming’ [48] or constructive
type theory [33] setting, carefully distinguishing between terminating functions
from algebraic datatypes and productive functions to coalgebraic codatatypes
(in which case the constructions are technically no longer ‘free’ monads), or
perhaps it really requires a shift from SET to CPO.

But none of these extensions will help when it comes to dealing with con-
tinuous probability distributions—say, a uniform choice among reals in the unit
interval. Any particular real result will have probability zero; if represented us-
ing the free monad DistT , all the leaves will have to be infinitely deep in the
tree, and no useful computation can be done. There is no fundamental reason
why one cannot deal with such distributions; after all, continuous probability
distributions form a monad too [27, 11, 20]. But it requires an approach based
on measure theory and Lebesgue integrals rather than point masses: distribu-
tions must be represented as mappings from measurable sets of outcomes to
probabilities, rather than from individual outcomes. We leave this for future
work.

For both of these reasons, possible nontermination and continuous distribu-
tions, the lightweight, embedded approach of unifying the ‘syntactic’ program-
ming notation with the ‘semantic’ reasoning framework—which is one of the
great appeals of functional programming—has its limitations. Morgan’s proba-
bilistic guarded command language approach, with a clearer separation between
the syntax and the semantics, pays the cost of two distinct domains of discourse,
but does not suffer from the same limitation.

9.3 Lawvere theories

The axiomatic approach to reasoning about effectful functional programs that
we have used here derives from our earlier paper [10]. It should come as no
surprise that algebraic theories—consisting of a signature for the operations,
together with laws that the operations are required to satisfy, but abstracting
away from specific models of the theory—are convenient for equational reason-
ing about programs; after all, algebraic specifications have long been espoused
as a useful technique for isloating the interface of a module from its possible
implementations, separating the concerns of the module provider and consumer
[7].

What did come as a surprise, at least to us, was that computational effects
such as nondeterminism and probabilistic choice are amenable to algebraic spec-
ifications. But history [19] tells us that this is only to be expected: algebraic
theories were introduced in Lawvere’s PhD thesis [26] as a category-theoretic
formulation of universal algebra; Linton [28] showed the equivalence between
such ‘Lawvere theories’ and monads (every Lawvere theory has a category of
models that is isomorphic to the category of algebras of some monad, unique up
to isomorphism), which arise as adjoint pairs of functors [8, 22]; and Moggi [36]
and Wadler [54] showed that monads are precisely what is needed to encapsulate
effects in pure functional languages. Indeed, this is precisely how impure effects
are implemented in a pure functional programming language such as Haskell:
pure evaluation is used to construct a term in the algebraic theory, which is sub-
sequently interpreted—with possible side-effects—by the impure run-time sys-
tem. In some ways, the algebraic theory approach to effects is more appealing
than the monadic approach, since it places the additional operations and their
properties front and centre; nevertheless, monads and Haskell’s do notation do
provide a rather elegant programming notation.

On the other hand, not all the additional operations we have discussed tech-
nically fit into the algebraic theory framework. Specifically, the bind operator
>>= should distribute over every such operation [42], as for example it does over
probabilistic choice:

do {x ← (p / w . q) ; k x } = do {x ← p ; k x } / w . do {x ← q ; k x }

But as we saw in Section 7, bind does not distribute over catch. Plotkin and
Pretnar [43] call operations like catch ‘effect handlers’; they need a separate
treatment.

